Fundação CECIERJ - Vice Presidência de Educação Superior a Distância

Curso de Tecnologia em Sistemas de Computação – UFF
Disciplina: Projeto e Desenvolvimento de Algoritmos
AP2 1° semestre de 2005.
Data 23/07/2005

Nome –

Assinatura –

Observações:

1. Prova sem consulta.
2. Coloque seu nome e assinatura na folha das questões e na folha de respostas.
3. Ao final da prova devolva as folhas de questões e as de respostas.
4. Todas as respostas devem ser transcritas na folha de respostas. As respostas na folha de questões não serão corrigidas.

As questões de números 1 a 3 devem ser resolvidas usando vetores

1ª questão (2.0 pontos)

O guichê de pedágio de uma rodovia possui um equipamento que registra diariamente a quantidade de carros que ali passaram. Faça um algoritmo para ler cada registro do mês de setembro (não necessariamente em ordem) e informar qual o maior volume de carros que passaram e em qual dia ele ocorreu. A entrada de dados termina depois que o movimento correspondente a todos os dias foi digitado.

Exemplo de uma entrada possível:

```
1     12025
23    11453
17    08545
5     15467
...
```

A saída gerada pelo programa deve ser:

```
O maior volume ocorreu no dia 5 e foi de 15467 carros
```

```
CONSTANTES
   DIM = 30
início
{entrada de dados}
   para i ← 1 até DIM faça
      leia idx
      leia movimento[idx]
   próximo i
{processamento e saída}
   maior ← movimento[1]
   onde ← 1
```

```
      para i ← 2 até DIM faça
         se (movimento[i] > maior) então
            maior ← movimento[i]
            onde ← i
         fim se
      próximo i

      imprima 'O maior volume ocorreu no dia ', onde, ' e foi de '
              + maior + ' carros'
fim
```

2ª questão (2.0 pontos)
Faça um algoritmo para ler dois preços de 15 produtos de uma cesta básica (anotados no início e no fim de uma semana) e imprimir uma listagem com o preço médio de cada produto.
Exemplo de uma entrada possível:

```
9.00        9.00
1.43        1.57
6.30        6.50
...
```

A saída gerada pelo programa deve ser:

```
Preço1      Preço2      Média
9.00        9.00        9.00
1.43        1.57        1.50
6.30        6.50        6.40
...
```

```
CONSTANTES
   DIM = 15
início
{entrada de dados}
   para i ← 1 até DIM faça
      leia inicio[i]
      leia fim[i]
   próximo i
{processamento e saída}
   imprima 'Preço1    Preço2    Média'
   para i ← 1 até DIM faça
      imprima inicio[i], '    ', fim[i], '    ',
              (inicio[i]+fim[i])/2.0)
   próximo i
fim
```

3ª questão (2.0 pontos)
Escreva um algoritmo que simule 6000 lançamentos de um dado e conte o número de ocorrências de cada uma das faces. Suponha que você dispõe de uma função *lançaDado* que retorna um valor entre 1 e 6.
Exemplo de uma saída possível para o programa:

```
1     973
2     981
3     1017
4     978
```

```
5     1046
6     1005
```

```
CONSTANTES
   DIM = 6000
início
   para i ← 1 até DIM faça
      face ← lançaDado()
      conta[face] ← conta[face] + 1
   próximo i

   para i ← 1 até 6 faça
      imprima conta[i]
   próximo i
fim
```

4ª questão (2.0 pontos)
Uma mercearia aplica uma política de descontos aplicada às compras à vista. Existem três estratégias de descontos:

Desconto fixo: R$ 2,00 para compras entre R$ 10,00 e R$ 50,00 e R$ 5,00 para compras acima deste valor
Desconto proporcional: 5% do valor da compra
Desconto por idade: Um desconto adicional de 10% para os clientes com mais de 65 anos. Este desconto é aplicado ao valor já reduzido pelos descontos anteriores.

Escreva um algoritmo que calcule o melhor preço para o cliente para uma série de compras. A entrada de dados termina quando o valor de uma compra é menor do que R$ 0,01.
Observação: Cada uma das estratégias de desconto deve ser calculada por uma função diferente. Sugestão: crie as funções *descontoFixo*, *descontoProporcional* e *descontoIdade*.
Exemplo:

```
valor da compra: 5
idade: 20
valor a pagar: 4.75
valor da compra: 20
idade: 20
valor a pagar: 18.0
valor da compra: 20
idade: 75
valor a pagar: 16.2
valor da compra: 100
idade: 35
valor a pagar: 95.0
valor da compra: 0
fim do programa
```

Observação: Foram mostrados em negrito os valores digitados pelo usuário.

```
programa questao4
inicio
   acabou ← falso
   enquanto não acabou faça
      imprima 'valor da compra: '
      leia compra
      se (compra < 0.01) então
         acabou ← verdadeiro
      senão
         imprima 'idade: '
         leia idade
         valorAPagar ← compra – max(descontoFixo(compra),
                       descontoProporcional(compra))
         imprima 'valor a pagar: ', valorAPagar –
                  descontoIdade(idade, valorAPagar)
      fim se
   fim enquanto
fim

função descontoFixo(entradas: valor)
início
   se valor > 50 então
      resultado ← 5
   senão
      se valor >= 10 então
         resultado ← 2
      senão
         resultado ← 0
      fim se
   fim se
fim

função descontoProporcional(entradas: valor)
início
   resultado ← 0.05*valor
fim

função descontoIdade(entradas: idade, valor)
início
   se idade > 65 então
      resultado ← 0.1*valor
   senão
      resultado ← 0
   fim se
fim

função max(entradas: valor1, valor2)
início
   se valor1 > valor2 então
      resultado ← valor1
   senão
      resultado ← valor2
   fim se
fim
```

5ª questão (2.0 pontos)
Determinar o valor da combinação de *m* elementos tomados *p* a *p*. O valor dessa combinação pode ser obtido através de:

$$C_m^p = \frac{m!}{p!(m-p)!}$$

Para facilitar a construção do seu algoritmo, escreva uma função *fatorial* que retorna o fatorial de um número inteiro passado como argumento. Lembre-se que o fatorial de um número inteiro *n* pode ser calculado como:
*n! = 1*2*3*...*n*

Exemplo:
```
m = 5
p = 3
C(5, 3) = 10
```

Observação: Foram mostrados em negrito os valores digitados pelo usuário.

```
programa questao5
início
   imprima 'm = '
   leia m
   imprima 'p = '
   leia p
   imprima 'C(', m, ', ', p, ') = ',
           fatorial(m)/fatorial(p)/fatorial(m-p)
fim

função fatorial(entradas: n)
início
   resultado ← 1
   para i ← 1 até n faça
      resultado ← resultado * i
   próximo i
fim
```